

# NAG Fortran Library Routine Document

## F08GPF (ZHPEVX)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F08GPF (ZHPEVX) computes selected eigenvalues and, optionally, eigenvectors of a complex  $n$  by  $n$  Hermitian matrix  $A$  in packed storage. Eigenvalues/vectors can be selected by specifying either a range of values or a range of indices for the desired eigenvalues.

### 2 Specification

```

SUBROUTINE F08GPF (JOBZ, RANGE, UPLO, N, AP, VL, VU, IL, IU, ABSTOL, M,
1                W, Z, LDZ, WORK, RWORK, IWORK, JFAIL, INFO)
    INTEGER          N, IL, IU, M, LDZ, IWORK(*), JFAIL(*), INFO
    double precision VL, VU, ABSTOL, W(*), RWORK(*)
    complex*16      AP(*), Z(LDZ,*), WORK(*)
    CHARACTER*1     JOBZ, RANGE, UPLO

```

The routine may be called by its LAPACK name *zhpevx*.

### 3 Description

The Hermitian matrix  $A$  is first reduced to real tridiagonal form, using unitary similarity transformations. The required eigenvalues and eigenvectors are then computed from the tridiagonal matrix; the method used depends upon whether all, or selected, eigenvalues and eigenvectors are required.

### 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: <http://www.netlib.org/lapack/lug>

Demmel J W and Kahan W (1990) Accurate singular values of bidiagonal matrices *SIAM J. Sci. Statist. Comput.* **11** 873–912

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

### 5 Parameters

- 1: JOBZ – CHARACTER\*1 *Input*  
*On entry:* if JOBZ = 'N', compute eigenvalues only.  
 If JOBZ = 'V', compute eigenvalues and eigenvectors.  
*Constraint:* JOBZ = 'N' or 'V'.
- 2: RANGE – CHARACTER\*1 *Input*  
*On entry:* if RANGE = 'A', all eigenvalues will be found.  
 If RANGE = 'V', all eigenvalues in the half-open interval (VL, VU] will be found.  
 If RANGE = 'I', the ILth to IUth eigenvalues will be found.

- 3: UPLO – CHARACTER\*1 *Input*  
*On entry:* if UPLO = 'U', the upper triangle of  $A$  is stored.  
 If UPLO = 'L', the lower triangle of  $A$  is stored.
- 4: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 5: AP(\*) – **complex\*16** array *Input/Output*  
**Note:** the dimension of the array AP must be at least  $\max(N \times (N + 1)/2)$ .  
*On entry:* the upper or lower triangle of the Hermitian matrix  $A$ , packed columnwise in a linear array. The  $j$ th column of  $A$  is stored in the array AP as follows:  
     if UPLO = 'U',  $AP(i + (j - 1) \times j/2) = a_{ij}$  for  $1 \leq i \leq j$ ;  
     if UPLO = 'L',  $AP(i + (j - 1) \times (2 \times n - j)/2) = a_{ij}$  for  $j \leq i \leq n$ .  
*On exit:* is overwritten by values generated during the reduction to tridiagonal form. If UPLO = 'U', the diagonal and first super-diagonal of the tridiagonal matrix  $T$  overwrite the corresponding elements of  $A$ , and if UPLO = 'L', the diagonal and first sub-diagonal of  $T$  overwrite the corresponding elements of  $A$ .
- 6: VL – **double precision** *Input*  
 7: VU – **double precision** *Input*  
*On entry:* if RANGE = 'V', the lower and upper bounds of the interval to be searched for eigenvalues.  
*Constraint:* VL < VU.  
 If RANGE = 'A' or 'T', VL and VU are not referenced.
- 8: IL – INTEGER *Input*  
 9: IU – INTEGER *Input*  
*On entry:* if RANGE = 'T', the indices (in ascending order) of the smallest and largest eigenvalues to be returned.  
 If RANGE = 'A' or 'V', IL and IU are not referenced.  
*Constraints:*  
     if  $N = 0$ , IL = 1 and IU = 0;  
     if  $N > 0$ ,  $1 \leq IL \leq IU \leq N$ .
- 10: ABSTOL – **double precision** *Input*  
*On entry:* the absolute error tolerance for the eigenvalues. An approximate eigenvalue is accepted as converged when it is determined to lie in an interval  $[a, b]$  of width less than or equal to  

$$ABSTOL + \epsilon \max(|a|, |b|),$$
 where  $\epsilon$  is the **machine precision**. If ABSTOL is less than or equal to zero, then  $\epsilon \|T\|_1$  will be used in its place, where  $T$  is the tridiagonal matrix obtained by reducing  $A$  to tridiagonal form. Eigenvalues will be computed most accurately when ABSTOL is set to twice the underflow threshold  $2 \times X02AMF()$ , not zero. If this routine returns with INFO > 0, indicating that some eigenvectors did not converge, try setting ABSTOL to  $2 \times X02AMF()$ . See Demmel and Kahan (1990).
- 11: M – INTEGER *Output*  
*On exit:* the total number of eigenvalues found.  
 If RANGE = 'A', M = N.

If RANGE = 'I',  $M = IU - IL + 1$ .

Constraint:  $0 \leq M \leq N$ .

12:  $W(*)$  – *double precision* array *Output*

**Note:** the dimension of the array  $W$  must be at least  $\max(1, N)$ .

*On exit:* if INFO = 0, the selected eigenvalues in ascending order.

13:  $Z(LDZ,*)$  – *complex\*16* array *Output*

**Note:** the second dimension of the array  $Z$  must be at least  $\max(1, M)$ .

*On exit:* if JOBZ = 'V', then if INFO = 0, the first  $m$  columns of  $Z$  contain the orthonormal eigenvectors of the matrix  $A$  corresponding to the selected eigenvalues, with the  $i$ th column of  $Z$  holding the eigenvector associated with  $W(i)$ .

If an eigenvector fails to converge, then that column of  $Z$  contains the latest approximation to the eigenvector, and the index of the eigenvector is returned in JFAIL.

If JOBZ = 'N',  $Z$  is not referenced.

**Note:** the user must ensure that at least  $\max(1, M)$  columns are supplied in the array  $Z$ ; if RANGE = 'V', the exact value of  $M$  is not known in advance and an upper bound must be used.

14: LDZ – INTEGER *Input*

*On entry:* the first dimension of the array  $Z$  as declared in the (sub)program from which F08GPF (ZHPEVX) is called.

Constraints:

if JOBZ = 'V',  $LDZ \geq \max(1, N)$ ;  
 $LDZ \geq 1$  otherwise.

15: WORK(\*) – *complex\*16* array *Workspace*

**Note:** the dimension of the array WORK must be at least  $\max(1, 2 \times N)$ .

16: RWORK(\*) – *double precision* array *Workspace*

**Note:** the dimension of the array RWORK must be at least  $\max(1, 7 \times N)$ .

17: IWORK(\*) – INTEGER array *Workspace*

**Note:** the dimension of the array IWORK must be at least  $\max(1, 5 \times N)$ .

18: JFAIL(\*) – INTEGER array *Output*

**Note:** the dimension of the array JFAIL must be at least  $\max(1, N)$ .

*On exit:* if JOBZ = 'V', then if INFO = 0, the first  $M$  elements of JFAIL are zero. If INFO > 0, JFAIL contains the indices of the eigenvectors that failed to converge.

If JOBZ = 'N', JFAIL is not referenced.

19: INFO – INTEGER *Output*

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO =  $-i$ , the  $i$ th argument had an illegal value.

INFO > 0

If INFO =  $i$ , then  $i$  eigenvectors failed to converge. Their indices are stored in array JFAIL. Please see ABSTOL.

## 7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix  $(A + E)$ , where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and  $\epsilon$  is the *machine precision*. See Section 4.7 of Anderson *et al.* (1999) for further details.

## 8 Further Comments

The total number of floating point operations is proportional to  $n^3$ .

The real analogue of this routine is F08GBF (DSPEVX).

## 9 Example

To find the eigenvalues in the half-open interval  $(-2, 2]$ , and the corresponding eigenvectors, of the Hermitian matrix

$$A = \begin{pmatrix} 1 & 2 - i & 3 - i & 4 - i \\ 2 + i & 2 & 3 - 2i & 4 - 2i \\ 3 + i & 3 + 2i & 3 & 4 - 3i \\ 4 + i & 4 + 2i & 4 + 3i & 4 \end{pmatrix}.$$

### 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F08GPF Example Program Text
*      Mark 21 Release. NAG Copyright 2004.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
INTEGER          NMAX, MMAX
PARAMETER       (NMAX=10,MMAX=5)
INTEGER          LDZ
PARAMETER       (LDZ=NMAX)
CHARACTER       UPLO
PARAMETER       (UPLO='U')
DOUBLE PRECISION ZERO
PARAMETER       (ZERO=0.0D+0)
*      .. Local Scalars ..
DOUBLE PRECISION ABSTOL, VL, VU
INTEGER          I, IFAIL, IL, INFO, IU, J, M, N
*      .. Local Arrays ..
COMPLEX *16      AP((NMAX*(NMAX+1))/2), WORK(2*NMAX), Z(LDZ,MMAX)
DOUBLE PRECISION RWORK(7*NMAX), W(NMAX)
INTEGER          INDEX(NMAX), IWORK(5*NMAX)
*      .. External Subroutines ..
EXTERNAL        X04DAF, ZHPEVX
*      .. Executable Statements ..
WRITE (NOUT,*) 'F08GPF Example Program Results'
WRITE (NOUT,*)
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN
*
*          Read the lower and upper bounds of the interval to be searched,
*          and read the upper or lower triangular part of the matrix A
```

```

*      from data file
*
      READ (NIN,*) VL, VU
      IF (UPLO.EQ.'U') THEN
        READ (NIN,*) ((AP(I+(J*(J-1))/2),J=I,N),I=1,N)
      ELSE IF (UPLO.EQ.'L') THEN
        READ (NIN,*) ((AP(I+((2*N-J)*(J-1))/2),J=1,I),I=1,N)
      END IF
*
*      Set the absolute error tolerance for eigenvalues. With ABSTOL
*      set to zero, the default value is used instead
*
      ABSTOL = ZERO
*
*      Solve the Hermitian eigenvalue problem
*
      CALL ZHPEVX('Vectors','Values in range',UPLO,N,AP,VL,VU,IL,IU,
+             ABSTOL,M,W,Z,LDZ,WORK,RWORK,IWORK,INDEX,INFO)
*
      IF (INFO.GE.0 .AND. M.LE.MMAX) THEN
*
*      Print solution
*
        WRITE (NOUT,99999) 'Number of eigenvalues found =', M
        WRITE (NOUT,*)
        WRITE (NOUT,*) 'Eigenvalues'
        WRITE (NOUT,99998) (W(J),J=1,M)
*
        IFAIL = 0
        CALL X04DAF('General',' ',N,M,Z,LDZ,'Selected eigenvectors',
+             IFAIL)
        IF (INFO.GT.0) THEN
          WRITE (NOUT,99999)
+             'INFO eigenvectors failed to converge, INFO =', INFO
          WRITE (NOUT,*)
+             'Indices of eigenvectors that did not converge'
          WRITE (NOUT,99997) (INDEX(J),J=1,M)
        END IF
        ELSE IF (M.GT.MMAX) THEN
          WRITE (NOUT,99996) 'M greater than MMAX, M =', M,
+             ', MMAX =', MMAX
        ELSE
          WRITE (NOUT,99999) 'Failure in ZHPEVX. INFO =', INFO
        END IF
      ELSE
        WRITE (NOUT,*)
        WRITE (NOUT,*) 'NMAX too small'
      END IF
      STOP
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (3X,(8F8.4))
99997 FORMAT (3X,(8I8))
99996 FORMAT (1X,A,I5,A,I5)
      END

```

## 9.2 Program Data

F08GPF Example Program Data

```

      4                               :Value of N
     -2.0          2.0                :Values of VL and VU

(1.0, 0.0) (2.0, -1.0) (3.0, -1.0) (4.0, -1.0)
           (2.0, 0.0) (3.0, -2.0) (4.0, -2.0)
                               (3.0, 0.0) (4.0, -3.0)
                               (4.0, 0.0) :End of matrix A

```

### 9.3 Program Results

F08GPF Example Program Results

Number of eigenvalues found = 2

Eigenvalues

-0.6886 1.1412

Selected eigenvectors

1 2

1 -0.3975 -0.3746  
0.5105 -0.2414

2 0.3953 0.2895  
-0.3238 -0.4917

3 -0.4309 0.3768  
0.0383 0.3994

4 0.3648 -0.4175  
0.0000 0.0000

---